
Abstract 
This paper describes how the author used SAS/IntrNet to develop a simple intranet 
application to provide auditors and investigators at the State Auditor’s Office with 
information extracted from several databases containing personnel information data for 
the State of Texas. 
 
Introduction 
Sometimes auditors need information pertaining to the employment status or authorized 
salary of certain individuals, as of a particular date.  The source data for responding to 
such queries are contained in three different databases that are maintained by the State 
Comptroller’s Office.   Instead of executing a separate query against the personnel 
databases each time a data request is received, the SAO decided to create its own version 
of the essential tables every quarter, containing the most frequently-requested columns, 
and to use those data for data requests, as needed.    The SAO also decided to develop an 
intranet application which auditors could use, without having to solicit assistance from 
programmers, in order to create standardized reports or data files containing the 
personnel data that they need.  
  
Early in 2004, using SAS/IntrNet software, the State Auditor’s Office created an intranet 
application for creating standardized reports or data files containing financial summary 
and detail information – including expenditures, revenues, encumbrances, budgets, and 
vendor payments. [Note -- This was reported in a paper at the SCSUG 2004 conference.  
See “Suggestions for Further Reading”, at the end of this paper.]  The well-received 
“USAS Guide” was intended to be the first of several useful SAS/IntrNet applications 
developed for use by auditors and investigators.   Developing the USAS Guide provided 
a template for development of other web-based applications.  One year later, the second 
SAS/IntrNet application to be put into production was Personnel Information Reports.    
This paper will describe some of the “nuts and bolts” for building that simple application.  
 
Our Computing Environment -- Hardware and Software 
In November 2002, in order to create a facility capable of  analyzing large data files,  the 
State Auditor's Office established a SAS server system consisting of a Dell PowerEdge™ 
2650 server (with two Pentium® III Xeon™ processors) connected to a Dell 
PowerVault™ storage system.  The storage system consists of one PowerVault 660F 14-
bay fiber channel RAID array and one PowerVault 220S 14-bay SCSI RAID array, both 
in RAID 5 configuration.  The server is configured with Microsoft® Windows® 2000 
Server.  Initially, the SAS System software installed on our server was limited to Base 
SAS®, SAS/SHARE®, SAS/STAT®, SAS/GRAPH®, SAS/CONNECT®, and 
SAS/ACCESS® Interface to PC File Formats.   And in October 2003, we added  
SAS/IntrNet®, and SAS/ACCESS® Interface to Sybase®. 
 
The SAO’s Web server has the same basic specifications as the SAS Server, except that 
there is no external storage system attached to it.  It is a Dell PowerEdge 2650 server with 

dual Pentium III Xeon processors.  The internal hard drives are configured in RAID 5 
configuration.   It is running Windows 2000 Server with Internet Information Services.  
 
At the SAO, there are only five Systems Analysts who are knowledgeable about SAS 
programming.  There are a few other programmers, who are proficient with certain 
development tools besides SAS.  And there are many non-programmers, who possess 
varying levels of familiarity with using several application software packages.   All SAO 
employees know how to access information on our intranet.  Therefore, SAS/IntrNet 
applications seem to be an excellent method for delivering information to SAO 
employees, since using such applications requires only a web browser on the user’s 
machine, and their users don’t need to have any SAS knowledge either.  
 
The Personnel Data 
Data concerning employees of the State of Texas is distributed between three different 
personnel systems, each system containing information for different collections of state 
agencies and institutions.   Although one might suppose that a single personnel and 
payroll system would be desirable, for historical reasons Texas has three distinct systems, 
and each system has different capabilities.    
 
The Texas Comptroller of Public Accounts maintains three systems for personnel and 
payroll data concerning all employees of the State of Texas.  These are: 

HRIS - Human Resources Information System, 
USPS - Uniform Statewide Payroll/Personnel System, and 
SPRS - Standardized Payroll/Personnel Reporting System. 

Some types of information are common to all three systems (for example, employee 
names, social security numbers, gender, race, birthdates, and job titles), but the column 
names and other attributes for similar data items often differ from one system to the next.   
USPS contains the greatest amount of useful information, as well as several optional 
human resources administration modules.  HRIS cannot generate a payroll, and it also 
lacks some duration-of-service measures.  SPRS does include amounts paid as 
compensation, but it contains only essential personnel data.      
 
The data for all three personnel systems are stored in DB2 databases on the Comptroller’s 
mainframe computer.  Unfortunately, that agency does not have SAS Integration 
Technologies, so we are unable to take advantage of the interoperability and easy 
information distribution that would be available using that SAS System component.   
Except during transitional periods, personnel information for any particular agency is 
stored in one, and only one, of the three personnel systems.  Therefore, to obtain 
historical information concerning the current or former employees of most agencies, one 
would have to look in different systems for various intervals of dates.   
 
About one month after the last day of a fiscal quarter, the most frequently-requested 
personnel information for all state employees as of the last day of the preceding fiscal 
quarter, and for all employee terminations and other personnel actions during the 
preceding fiscal quarter are extracted from the HRIS, USPS, and SPRS databases on the 
Comptroller’s mainframe computer.  These data are downloaded to the SAO network 
environment, and are saved in a SAS Data Library on the SAS Server.  This is 
accomplished using SQL Pass-Through, DATA step programming, certain basic SAS 

procedures, and FTP.  Here is a peek at the PROC CONTENTS output for some of the 
members from that SAS Data Library. 

An Overview of the Personnel Information Reports Application 
  
Here is a conceptual diagram which describes the interaction of the operational 
components that are used by the Personnel Information Reports intranet application at the 
SAO: 

 
 
The Personnel Information Reports intranet application is made possible by SAS/IntrNet 
software, using Compute Services with the Application Dispatcher.  Application 
Dispatcher is composed of two pieces: (1) the Application Broker, which is a Common 
Gateway Interface (CGI) program that resides on the web server, and communicates 
between the browser and the SAS Application Server, and (2) the SAS/IntrNet compute 
server, which is a SAS program on the SAS Server.   Since the SAO’s  has SAS/SHARE 
installed on the SAS Server, it also would be possible for the SAO to develop 
SAS/IntrNet applications that make use of Data Services through execution of the 
htmSQL CGI program on the Web server; however, the Personnel Information Reports 
application does not use htmSQL.  [However, the USAS Guide does!]    
 
The Personnel Information Reports application is available to all SAO employees by 
following a straightforward path from any intranet page.   In the application, requestors 
are prompted to make selections from a web page, and ultimately from one of three 
HTML forms.  The parameters from the HTML form are passed to a SAS program as 
values of macro variables, and the program executes a query and generates HTML 
output, which is sent back to the requestor.  All of the HTML documents, are kept in a 
single folder on the web server.  They are retrieved from the web server by referencing 
their URLs.  The SAS programs and data are stored on the SAS Server. 

The web pages pertaining to the data requests were constructed using Notepad and 
Macromedia Dreamweaver MX software.   Like all SAO internet/intranet applications, 
the Personnel Information Reports application conforms to agency standards for web 
applications.  For example, a particular style sheet is used repeatedly by all of the HTML 
files which generate data requests.  It provides a standardized header, as well as some 
links to certain other SAO web pages.   There also are standards which pertain to the 
coding and formatting of web content.  For example, we are prohibited from using 
frames, and also from using certain colors. 
    
Operating the Personnel Information Reports Application 
Here is a portion of the main request page for the application: 
 

 
 
 
The main request page presents the user with links to three data request pages.  
Let us suppose that the user is interested in reports containing an employee listing for a 
specific agency and as-of date.  So, for example, if the user clicked on the link for 
“Employee Listing Reports”, then the following HTML form would be returned: 

The user would make his/her selections from the drop-down boxes, and then would click 
on the “Generate Report” button.  At this point, the important details for the specific data 
request would be sent from the Web browser to the SAS Application Broker CGI 
program, which runs on the Web server.  Then, the Application Broker would access the 
SAS Application Server, and would pass the parameters for the user’s particular 
selections to the appropriate SAS programs.   
 
In the Personnel Information Reports application, all data requests are handled 
dynamically, which means that the application does not include a collection of pre-
prepared reports, covering all possible requests, stored as HTML documents, which 
would be displayed whenever one of them was requested.  The application could have 
been designed to work that way, but it wasn’t -- because I didn’t want to have to manage 
a collection of separate, static reports.   Instead, each data request results in the execution 
of SAS programs which extract the data which are pertinent to the request from 
appropriate SAS data sets, and then create HTML web content using the SAS Output 
Delivery System (ODS).     
 
If I had decided to build this application using static web pages, then accommodating just 
five fiscal years of data for the Personnel Information Reports application would have 
required more than 113,400 static web pages! 

Let us suppose that the user made the following selections for his/her data request: 
“Agency 352 – Texas Bond Review Board”, employees as of “30Nov2004”, sorted 
according to “Employee Name”, and with the output to be formatted as HTML. The next 
display that the user would see would be the following web page: 
 

 
 
 
Notice the URL in the “Address” line near the top of the Internet Explorer window.  

 
This line tells us that the SAS program launched on the SAS Server is empl_list.sas. The 
URL also contains the values of parameters that pertain to the user’s specific selections 
for this query, “agy”, “reqdat”, “sortvar”, and “outfmt”.   In the request, agy=352 
identifies Agency #352, which is the Texas Bond Review Board, the “as-of-date” 
reqdat=30Nov2004, the sorting-variable sortvar=Name, and the output-format 
outfmt=html.  Those parameters were inserted into the particular SAS program, 
empl_list.sas, as values of macro variables, before it was executed.   The program 
generated a standardized report containing data items have been found to be useful for 
most data requests of the same type.  The program used ODS to write the output to 
“_webout “, which specifies that the results are to be sent directly to the Web browser.       
I could have hidden those value-pairs in the browser address line, by changing the 
METHOD=GET attribute to METHOD=POST in the FORM tag of the HTML that was 
sent to SAS by the web server.   

The SAS program checks the values of the parameters that are sent to it.  If  there are no 
user-specified values for agy, reqdat, or sortvar, then the following message is sent to the 
user. 
 

 
 
 
After the user submits a request, SAS extracts the data from the tables.  This generally 
doesn’t take too long.   After the data are selected, but just before the desired report is 
generated, SAS sends a message back to the browser to run some Javascript code that 
asks for the user to be patient while the report is being formatted.   For short queries, the 
user may never actually see the “Please Wait”-message.  But for longer-running requests 
the user will see the following screen, until SAS completes the report and sends it to the 
browser: 

I have demonstrated that the application can generate output reports as HTML.  If 
desired, the user could have selected an output file of comma-separated values (CSV), 
instead.  A CSV file is a presentation of a table as a series of ASCII text lines, in which 
each column value in the table row is separated from the value in the next column by a 
comma, and each table row is a new line.  CSV files can be read by various application 
programs, including Microsoft Excel, Microsoft Access, and ACL.   An example of how 
the Personnel Information Reports application can be used to generate CSV files follows.  
Notice that the name of the CSV file that is created for download includes elements 
which identify the type of request (empls, terms, or actions), the agency number, and the 
period.  

Besides generating data products concerning employee listings, terminations, and other 
personnel actions, the application also includes two utilities for doing name lookups, 
based upon user-specified name-fragments (a name-fragment can be any string of text, 
excluding commas, double-quotes, and other non-alphabetical characters): 

(a)  The “Agency Employee Name Lookups” tool generates a listing of all 
employees of the selected state agency and as-of date, whose names contain 
ANY of the specified name-fragments, and 

(b)  The “All Agencies – Employee Name Lookups” tool generates a listing of all 
employees of any state agency, as of the selected date, whose names contain 
ALL of the specified name-fragments.  

 
Here are the two name lookup HTML forms, plus a couple of examples of results of 
queries:  

Now, one might think that a user would only need one of those name lookup tools, but 
they each are a little bit different.  To begin with, "Agency Employee Name Lookups" 
only performs name-fragment searches within a single, user-specified agency for a 
particular date, while "All Agencies – Employee Name Lookups" performs name-
fragment searches across all agencies for the same set of dates; therefore, each tool casts 
its net into two very differently-sized pools.   Next, observe that while "Agency 
Employee Name Lookups" only does searches for a single agency, nevertheless its search 
criteria is much more “loose” than the criteria used in "All Agencies – Employee Name 
Lookups"  for determining matches across all state agencies.  In "Agency Employee 
Name Lookups", the user may provide from one to ten name-fragments, and the tool will 
find instances where one or several of the name-fragments occur.  In "All Agencies – 
Employee Name Lookups", the user may specify from two to ten name-fragments, and 
the tool will look for cases in which all of the specified name-fragments occur.  
Therefore, "Agency Employee Name Lookups" is a more exploratory utility, which 
someone could use to search for various combinations of the name-fragments.  This 
would be particularly useful, for example, in searches for name-fragments which might 
sound similar, but have different spellings.    However, to use "All Agencies – Employee 
Name Lookups", one must have a more precise idea of what he/she is looking for.   If not, 
the user could wind up with unintended results. 

From a technical point of view, the user selections submitted from either of the name 
lookup HTML forms generates name-value pairs whose parameters are inserted into the 
appropriate SAS program as values of macro variables.  The user’s selections for the 
name-fragments are passed to SAS as a single string of characters, which SAS separates 
into name-fragments, and then builds a subsetting WHERE clause using the CONTAINS 
operator.  (the actual code is included in the Appendix)  
 
Miscellaneous Remarks. 
During development and testing of the application, a very useful technique for debugging 
SAS programs invoked through the Application Dispatcher was to append the debug flag 
and associated parameter value “&_debug=131” to the URL.  This returns all values 
passed to the SAS Server, the SAS Log, and the total elapsed time.  

(followed by more pages of the SAS Log, . . . and then finally) 

The Personnel Information Reports intranet application does not include data analysis 
capabilities, and no customizable details pertaining to the data products themselves.  
Providing those functionalities would have required a more complicated application than 
the one that I developed.  However, our auditors prefer performing their own data 
analysis using other software tools (currently Microsoft Excel, Microsoft Access, and 
ACL).   Therefore, the Personnel Information Reports application seems to meet their 
needs very well.   It is easy to use, and it includes the most-commonly requested types of 
personnel information needed by auditors (employee listings, as well as terminations, 
acquisitions, promotions, demotions, merit increases, reclassifications, salary actions, and 
various other personnel actions, and name lookups).   Of course, not all of our audits 
involve personnel information. 
 
Like its predecessor, the USAS Guide, the Personnel Information Reports application is a 
“work in progress”, and it is anticipated that making improvements to it will be an 
ongoing task.  Additional modules under consideration for future development within the 
Personnel Information Reports application include payroll and leave accounting data.   

Conclusion 

• SAS/IntrNet software was used to create an intranet application for auditors and 
investigators (non-SAS programmers) for creating standardized data products from a 
database of personnel information. 

• This paper provided descriptions of the development and operation of the Personnel 
Information Reports intranet application.  

– Appendix – 
Selected HTML Code and SAS Programs  

Used by the Personnel Information Reports Intranet Application 
 

Here is the HTML file that is used for the application’s main page.  It is located on the 
intranet server. 
 

<?xml version="1.0" encoding="utf-8"?> 
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "DTD/xhtml1-transitional.dtd"> 
<HTML xmlns="http://www.w3.org/1999/xhtml" xml:lang="en" lang="en"> 
<HEAD> 
 <link rel="stylesheet" type="text/css" 
href="http://www.sao.state.tx.us/templates/style.css" /> 
 <link rel="stylesheet" type="text/css" href="/intranet/templates/intranet_style.css" 
/> 
 <CF_Intranet> 
 <cf_nocache> 
  <meta name="Description" content="SAO Information Systems Support Team" /> 
 <meta name="Keywords" content="Information Systems Support Team" /> 
 <title>Standard Requests for State Agency Personnel Information</title> 
 <!--- Set Page Contact & Review Cycle. If no Contact Link, comment out the <cfset 
ContactLink...> line. Review Cycle is Annually by default unless Reviewed variable is set 
here. ---> 
 <cfset Contact="Tom Winn"> 
 <cfset ContactLink="mailto:twinn@sao.state.tx.us"> 
<link rel="stylesheet" href="../../../Manuals/Empls/style_usas.css" type="text/css" /> 
<meta http-equiv="Content-Type" content="text/html; charset=utf-8" /> 
 
  
</HEAD> 
 
<BODY> 
 
<cfinclude template="#root#templates/master.html"> 
<!---MAIN CONTENT BEGINS------------------------------------------------------------> 
<div class="top">  <p class="section"><a href="../default.html">Audit Tools</a>: Personnel 
Information Reports:</p> 
  <H1><a name="LISTIndex" id="LISTIndex"></a>State Agency Personnel Information Requests</H1> 
</div> 
 
<p> You have reached the request page for obtaining personnel information pertaining to state 
agencies.  You may select one of the following: </p> 
<ul> 
   <li><a href="empl_list2.html">Employee Listing Reports</a></li> 
   <li><a href="terms.html">Employee Terminations Reports</a></li> 
   <li><a href="actions.html">Selected Personnel Actions Reports</a></li> 
   <li><a href="empl_lookup.html">Agency Employee Name Lookups</a></li> 
   <li><a href="empl_lookup_all_agys.html">All Agencies - Employee Name Lookups</a></li> 
</ul> 
 
</BODY> 
 
</HTML>   

 
 
Here is the HTML file that is used for soliciting the user-specified request parameters for 
the Employee Terminations Report, and for calling the SAS program that generates that 
report.  It also is located on the intranet server. 
    

<?xml version="1.0" encoding="utf-8"?> 
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "DTD/xhtml1-transitional.dtd"> 
<HTML xmlns="http://www.w3.org/1999/xhtml" xml:lang="en" lang="en"> 
<HEAD> 
 <link rel="stylesheet" type="text/css" 
href="http://www.sao.state.tx.us/templates/style.css" /> 
 <link rel="stylesheet" type="text/css" href="/intranet/templates/intranet_style.css" 
/> 
 <CF_Intranet> 
 <cf_nocache> 
  <meta name="Description" content="SAO Information Systems Support Team" /> 
 <meta name="Keywords" content="Information Systems Support Team" /> 
 <title>Make Selections for Employee Terminations Report</title> 
 <!--- Set Page Contact & Review Cycle. If no Contact Link, comment out the <cfset 
ContactLink...> line. Review Cycle is Annually by default unless Reviewed variable is set 
here. ---> 
 <cfset Contact="Tom Winn"> 
 <cfset ContactLink="mailto:twinn@sao.state.tx.us"> 

<link rel="stylesheet" href="../../../Manuals/Empls/style_usas.css" type="text/css" /> 
<meta http-equiv="Content-Type" content="text/html; charset=utf-8" /> 
 
  
<SCRIPT LANGUAGE="javascript"> 
function validate() { 
if (document.form1.agy.value == "") 
{ alert('Agency Number is Required');  
  document,form1.agy.focus(); 
   return false;  } 
   return true; 
  }   
 
</SCRIPT>  
 
 
</HEAD> 
 
<BODY> 
 
<cfinclude template="#root#templates/master.html"> 
<!---MAIN CONTENT BEGINS------------------------------------------------------------> 
<div class="top">  <p class="section"><a href="../default.html">Audit Tools</a>: <a 
href="default.html">Personnel Information Reports</a>: Employee Terminations Reports:</p> 
  <H1><a name="LISTIndex" id="LISTIndex"></a>Employee Terminations Request Form</H1> 
</div> 
Social Security Numbers are confidential information.  If you have a legitimate need for a 
listing that includes SSNs, then contact Olin Davis or Tom Winn. 
<HR /> 
<FORM ACTION="http://saointranet/sas/sas-cgi/broker.exe?" METHOD=get> 
<input type="hidden" name="_service" value="SAO" /> 
<input type="hidden" name="format" value="W" /> 
<input type="hidden" name="_program" value="saoprog.terms_list.sas" /> 
 
Make your selections for a report listing all employee terminations of a particular agency, 
during a specified period: <BR /> 
<P> <STRONG>1. Select Agency:&nbsp;</STRONG><BR /> 
<cfinclude template="Style_empls_agys.cfm"> 
 
</p><P><STRONG>2. Select period:&nbsp; </STRONG><BR /> 
<SELECT NAME=period SIZE=4 style="background-color: rgb(255,255,196); color: rgb(0,0,128); 
font-weight: bold; "> 
        <option selected="selected" value="x"> Select a period from the listbox below</option> 
        <OPTION VALUE=FY2000>FY2000</OPTION> 
        <OPTION VALUE=FY2001>FY2001</OPTION> 
        <OPTION VALUE=FY2002>FY2002</OPTION> 
        <OPTION VALUE=FY2003>FY2003</OPTION> 
        <OPTION VALUE=FY2004>FY2004</OPTION> 
        <OPTION VALUE=Q1FY2005>Q1FY2005</OPTION> 
        
</SELECT> 
</p><P><STRONG>3. Sort by:&nbsp; </STRONG><BR /> 
<SELECT NAME=sortvar SIZE=3 style="background-color: rgb(255,255,196); color: rgb(0,0,128); 
font-weight: bold; "> 
        <option selected="selected" value="y"> Select a sorting variable from the listbox 
below</option> 
        <OPTION VALUE=Name>Employee Name</OPTION> 
        <OPTION VALUE=Eff_Date1>Effective Date of Termination</OPTION> 
        <OPTION VALUE=HRIS_Reason>Reason Code</OPTION> 
</SELECT> 
</p><P><STRONG>4. Output format:&nbsp; </STRONG><BR /> 
<INPUT TYPE=radio NAME=outfmt VALUE=html CHECKED />HTML 
<INPUT TYPE=radio NAME=outfmt VALUE=csv />Comma-Separated Values 
<BR /> 
 
<STRONG>5. Generate the report or reset the  
values as necessary:&nbsp;   
<INPUT NAME="submit" TYPE="submit" VALUE="Generate Report" />&nbsp;&nbsp;  
<INPUT NAME="reset" TYPE="reset" VALUE="Reset Values" /> 
</STRONG></P> 
<HR /> 
<P><STRONG>For questions or comments please contact</STRONG>  
<BR /><STRONG>Tom Winn, Information Systems Support Team</STRONG><BR /> 
&nbsp;&nbsp;&nbsp;&nbsp;phone: (512) 936-9735<BR /> 
&nbsp;&nbsp;&nbsp;&nbsp;e-mail: <A 
HREF="mailto:twinn@sao.state.tx.us">twinn@sao.state.tx.us</A></P>   
</FORM> 
</BODY> 
 
</HTML> 

Here is the SAS program for the Employee Terminations Report.  It extracts the data and 
creates the report, and also sends the “We’re sorry …” and “Please wait …” messages to 
the browser, if they are pertinent.  It is located on the SAS Application Server. 
 

*******************************************************************************************; 
********************************   Program terms_list.sas   *******************************; 
*******************************************************************************************; 
***  This is the Dispatcher Application program for the "Employee Terminations Report". ***; 
***  It is invoked by the HTML Form, terms_list_form1.html, and utilizes user-specified ***; 
***  selections for agency number, period, and sorting variable.                        ***; 
***                                                 03-14-2005   Tom Winn   936-9735    ***; 
*******************************************************************************************; 
 
libname empls 'G:\Empls'; 
options obs=max; 
 
options symbolgen; 
 
%macro drvterms; 
   %global agy period sortvar outfmt; 
   %if &agy=z or &period=x or &sortvar=y %then %do; 
         %whoops 
   %end; 
   %else %do; 
      data pickem; 
         set empls.terms_&period; 
         if Agency = "&agy"; 
         if         HRIS_Reason = '051' then Reason = 'Inadequate salary                            '; 
            else if HRIS_Reason = '052' then Reason = 'Lack of opportunity for advancement          '; 
            else if HRIS_Reason = '053' then Reason = 'Dissatisfaction with supervisor or coworkers '; 
            else if HRIS_Reason = '054' then Reason = 'Dissatisfaction with travel                  '; 
            else if HRIS_Reason = '055' then Reason = 'Working hours                                '; 
            else if HRIS_Reason = '056' then Reason = 'Dislike/Unsuitability for assigned duties    '; 
            else if HRIS_Reason = '057' then Reason = 'Resignation in lieu of involuntary separation'; 
            else if HRIS_Reason = '058' then Reason = 'Personal reasons not related to job          '; 
            else if HRIS_Reason = '059' then Reason = 'Reasons unknown                              '; 
            else if HRIS_Reason = '060' then Reason = 'Voluntary termination                        '; 
            else if HRIS_Reason = '063' then Reason = 'Termination at will                          '; 
            else if HRIS_Reason = '064' then Reason = 'Reduction in force                           '; 
            else if HRIS_Reason = '065' then Reason = 'Transfer to another state agency/institution '; 
            else if HRIS_Reason = '067' then Reason = 'Dismissal for cause                          '; 
            else if HRIS_Reason = '068' then Reason = 'Retirement                                   '; 
            else if HRIS_Reason = '069' then Reason = 'Death                                        '; 
            else                             Reason = 'Other                                        '; 
          keep Agency Name Job_Class_cd Job_Description Hire_Date1 Eff_Date1  
               HRIS_Reason Reason Monthly_Salary; 
          run; 
 
      proc sort data=pickem; 
         by &sortvar; 
      run; 
   
      %if &outfmt=html %then %do; 
         %termhtm 
      %end;  
   %else %if &outfmt=csv %then %do; 
         %termcsv 
   %end; 
   %end; 
%mend drvterms; 
 
%macro whoops; 
   data _null_; 
      file _webout; 
   put 'Content-type: text/html'//; 
   put ; 
   put '<HTML>'; 
   put '<HEAD>'; 
   put '<TITLE> Selection Errors for Employee Terminations Request </TITLE>'; 
      put '</HEAD>'; 
      put '<BODY BGCOLOR="#FFFFFF">'; 
   put '<br /><br /><br />'; 
      put '<center><img src="I:/intranet/images/stop sign.bmp"><br />'; 
   put '<H2 ALIGN="center">'; 
   put '<FONT COLOR=red>'; 
   put '<P>'; 
   put 'We''re sorry . . .'; 
   put 'You must select values for Agency, period, and a sorting variable.'; 
   put '</H2>'; 
   put 'Please try again.'; 
   put '</FONT>'; 
   put '</BODY>'; 
   put '</HTML>'; 
 run; 
%mend whoops; 
 
%macro termhtm; 
   %global prtline; 

   %if &sortvar=Name %then %let prtline=Name Job_Class_cd Job_Description Hire_Date1 Eff_Date1 HRIS_Reason 
Reason Monthly_Salary; 
   %else %if &sortvar=Eff_Date1 %then %let prtline=Eff_Date1 Name Job_Class_cd Job_Description Hire_Date1 
HRIS_Reason Reason Monthly_Salary; 
   %else %if &sortvar=HRIS_Reason %then %let prtline=HRIS_Reason Reason Eff_Date1 Name Job_Class_cd 
Job_Description Hire_Date1 Monthly_Salary;  
 
   ods listing close; 
   ods html body=_webout (dynamic title="Employee Terminations Report") rs=none; 
 
   data _null_;   
      file _webout; 
      put '<div id="pleasewait" style="position:absolute; height:100%; width:100%;">'; 
      put '<br /><br /><br /><br /><br />'; 
      put '<center><img src="I:/intranet/images/notice.bmp"><br /><br /><br />'; 
      put '<b><font face="Arial" color="#000099" size="3">'; 
      put 'Your Request is Being Processed, Please Wait . . .</font></center></b><br /><br />'; 
   put '</div>'; 
      put '<div id="output" style="visibility:hidden">'; 
   run; 
 
   proc print data=pickem label n noobs; 
      var &prtline; 
   format Hire_Date1 Eff_Date1 date9. 
             Monthly_Salary dollar12.2; 
      label Job_Class_cd='Job Classification' 
            Job_Description='Job Title' 
            Hire_Date1='Date of Last Employment' 
         Eff_Date1='Effective Date of Termination' 
            HRIS_Reason='Reason Code' 
            Monthly_Salary = 'Amount of Monthly Salary'; 
      title1 "Employee Terminations for Agency #&agy, during &period,"; 
      title2 "Sorted According to Values of &sortvar"; 
   run; 
   
   data _null_; 
      file _webout; 
      put '</div>'; 
      put '<script language="JavaScript1.2">'; 
      put '   document.getElementById("pleasewait").style.visibility   = "hidden"'; 
      put '   document.getElementById("output").style.visibility       = "visible"'; 
      put '</script>'; 
   run; 
 
   ods html close; 
   ods listing; 
%mend termhtm; 
 
%macro termcsv; 
   %global prtline; 
   %if &sortvar=Name %then %let prtline=Name Job_Class_cd Job_Description Hire_Date1 Eff_Date1 HRIS_Reason 
Reason Monthly_Salary; 
   %else %if &sortvar=Eff_Date1 %then %let prtline=Eff_Date1 Name Job_Class_cd Job_Description Hire_Date1 
HRIS_Reason Reason Monthly_Salary; 
   %else %if &sortvar=HRIS_Reason %then %let prtline=HRIS_Reason Reason Eff_Date1 Name Job_Class_cd 
Job_Description Hire_Date1 Monthly_Salary;  
 
   ods html close; 
   ods listing; 
 
   data term2; 
      retain &prtline; 
      set pickem; 
      keep &prtline;      
      label Job_Class_cd='Job Classification' 
            Job_Description='Job Title' 
         Hire_Date1='Date of Last Employment' 
         Eff_Date1='Effective Date of Termination' 
            HRIS_Reason='Reason Code' 
            Monthly_Salary = 'Amount of Monthly Salary'; 
   run; 
   
   %ds2csv(data=term2,conttype=y,contdisp=y, 
        savefile=terms_&agy._&period..csv, 
        csvfref=_webout,runmode=s,openmode=replace); 
 
%mend termcsv; 
 
%drvterms 
 
 
  
 

Here is the SAS program for the Employee Name Lookups tool.  It extracts the data and 
creates the report, and also sends the “We’re sorry …” and “Please wait …” messages to 
the browser, if they are pertinent.  It is located on the SAS Application Server. 
 

*******************************************************************************************; 
****************************   Program empl_name_lookup.sas   *****************************; 
*******************************************************************************************; 
***  This is the Dispatcher Application program for the "Employee Name Lookup Report".  ***; 

***  It is invoked by the HTML Form, empl_lookup.html, and utilizes user-specified      ***; 
***  selections for agency number, one or more name-items, and as-of date.              ***; 
***                                                 06-28-2005   Tom Winn   936-9735    ***; 
*******************************************************************************************; 
 
libname empls 'G:\Empls'; 
options obs=max; 
options symbolgen; 
 
%macro drvempls; 
   %global agy namstrng reqdat nam_str; 
   %if &agy=z or "&namstrng"="Default text" or %scan(&namstrng,1)=  or &reqdat=x %then %do; 
         %whoops 
   %end; 
   %else %do; 
      %let nam_str = %qsysfunc(appsrv_unsafe(namstrng)); 
      data _null_; 
         namstrng = upcase("&namstrng"); 
         if namstrng ne '' then do; 
            call symput('namstrng',upcase(trim(left("&namstrng.")))); 
         end; 
      run; 
      %namstr 
   %emplhtm 
   %end;   
%mend drvempls; 
 
%macro whoops; 
   data _null_; 
      file _webout; 
   put 'Content-type: text/html'//; 
   put ; 
   put '<HTML>'; 
   put '<HEAD>'; 
   put '<TITLE> Selection Errors for Employee Name-Lookup Request </TITLE>'; 
      put '</HEAD>'; 
      put '<BODY BGCOLOR="#FFFFFF">'; 
   put '<br /><br /><br />'; 
      put '<center><img src="http://saointranet/intranet/images/stop sign.bmp"><br />'; 
   put '<H2 ALIGN="center">'; 
   put '<FONT COLOR=red>'; 
   put '<P>'; 
   put 'We''re sorry . . .'; 
   put 'You must select values for Agency, one or more name-items, and as-of date.'; 
   put '</H2>'; 
   put 'Please try again.'; 
   put '</FONT>'; 
   put '</BODY>'; 
   put '</HTML>'; 
 run; 
%mend whoops; 
 
%macro namstr; 
   %global nam01 nam02 nam03 nam04 nam05 nam06 nam07 nam08 nam09 nam10 whr_cond; 
   %let nam01 = %upcase(%scan(&namstrng,1)); 
   %let nam02 = %upcase(%scan(&namstrng,2)); 
   %let nam03 = %upcase(%scan(&namstrng,3)); 
   %let nam04 = %upcase(%scan(&namstrng,4)); 
   %let nam05 = %upcase(%scan(&namstrng,5)); 
   %let nam06 = %upcase(%scan(&namstrng,6)); 
   %let nam07 = %upcase(%scan(&namstrng,7)); 
   %let nam08 = %upcase(%scan(&namstrng,8)); 
   %let nam09 = %upcase(%scan(&namstrng,9)); 
   %let nam10 = %upcase(%scan(&namstrng,10)); 
   %if &nam10 ne        %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  
                                                   or compress(Name,"'") contains "&nam05"  
                                                   or compress(Name,"'") contains "&nam06" 
                                                   or compress(Name,"'") contains "&nam07"  
                                                   or compress(Name,"'") contains "&nam08"  
                                                   or compress(Name,"'") contains "&nam09"  
                                                   or compress(Name,"'") contains "&nam10"); 
   %else %if &nam09 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  
                                                   or compress(Name,"'") contains "&nam05"  
                                                   or compress(Name,"'") contains "&nam06" 
                                                   or compress(Name,"'") contains "&nam07"  
                                                   or compress(Name,"'") contains "&nam08"  
                                                   or compress(Name,"'") contains "&nam09"); 
   %else %if &nam08 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  
                                                   or compress(Name,"'") contains "&nam05"  
                                                   or compress(Name,"'") contains "&nam06" 
                                                   or compress(Name,"'") contains "&nam07"  
                                                   or compress(Name,"'") contains "&nam08");  
   %else %if &nam07 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  
                                                   or compress(Name,"'") contains "&nam05"  
                                                   or compress(Name,"'") contains "&nam06" 
                                                   or compress(Name,"'") contains "&nam07"); 
   %else %if &nam06 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  
                                                   or compress(Name,"'") contains "&nam05"  
                                                   or compress(Name,"'") contains "&nam06"); 
   %else %if &nam05 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04"  

                                                   or compress(Name,"'") contains "&nam05"); 
   %else %if &nam04 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"        
                                                   or compress(Name,"'") contains "&nam03"  
                                                   or compress(Name,"'") contains "&nam04");  
   %else %if &nam03 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02"  
                                                   or compress(Name,"'") contains "&nam03"); 
   %else %if &nam02 ne  %then %let whr_cond = and (compress(Name,"'") contains "&nam01"  
                                                   or compress(Name,"'") contains "&nam02");  
   %else %if &nam01 ne  %then %let whr_cond = and compress(Name,"'") contains "&nam01"; 
   %else %if &nam01 =   %then %let whr_cond = ; 
   run; 
%mend namstr; 
 
%macro emplhtm; 
   %global nums; 
   data pickem; 
      set empls.empls_&reqdat(keep=Agency Name Job_Class_cd Job_Description Hire_Date1 durstasvc 
Monthly_Salary); 
         where Agency = "&agy" &whr_cond; 
   run; 
 
   data _null_; 
      set pickem nobs=count; 
   call symput('nums', count); 
   run; 
 
   ods listing close; 
   ods html body=_webout (dynamic title="Employee Name-Lookup Report") rs=none; 
 
   data _null_;   
      file _webout; 
      put '<div id="pleasewait" style="position:absolute; height:100%; width:100%;">'; 
      put '<br /><br /><br /><br /><br />'; 
      put '<center><img src="http://saointranet/intranet/images/notice.bmp"><br /><br /><br />'; 
      put '<b><font face="Arial" color="#000099" size="3">'; 
      put 'Your Request is Being Processed, Please Wait . . .</font></center></b><br /><br />'; 
   put '</div>'; 
      put '<div id="output" style="visibility:hidden">'; 
   run; 
 
   %if &nums < 1 %then %do; 
       %nodata 
   %end; 
   %else %do; 
      %doit 
   %end; 
 
   data _null_; 
      file _webout; 
      put '</div>'; 
      put '<script language="JavaScript1.2">'; 
      put '   document.getElementById("pleasewait").style.visibility   = "hidden"'; 
      put '   document.getElementById("output").style.visibility       = "visible"'; 
      put '</script>'; 
   run; 
 
   ods html close; 
   ods listing; 
%mend emplhtm; 
 
%macro doit; 
   proc print data=pickem label n noobs; 
      var Name Job_Class_cd Job_Description Hire_Date1 durstasvc Monthly_Salary; 
   format durstasvc 6.3; 
      label Job_Class_cd='Job Classification' 
            Job_Description='Job Title' 
      Hire_Date1='Date of Current Employment' 
      durstasvc='Total Years of State Service' 
            Monthly_Salary='Monthly Salary Amount'; 
      title1 "Employee Name-Lookup for Agency #&agy, and Name-item(s)<&nam_str.>,"; 
      title2 "as of &reqdat"; 
   run; 
%mend doit; 
 
%macro nodata; 
   data _null_;   
      file _webout; 
      put '</div>'; 
      put '<div id="sorry" style="position:absolute; height:100%; width:100%;">'; 
      put '<br /><br /><br /><br /><br />'; 
      put '<center><img src="http://saointranet/intranet/images/notice.bmp"><br /><br /><br />'; 
      put '<b><font face="Arial" color="#000099" size="5">'; 
      put '<FONT COLOR=red>'; 
   put '<P>'; 
   put 'We''re sorry . . . <br /><br />'; 
   put 'No data were found using the specified selection criteria. <br />'; 
   put 'Please try again with different input values. <br />'; 
   put '</FONT></center>'; 
   put '</div>'; 
      put '<div id="output" style="visibility:hidden">'; 
   run; 
%mend nodata; 
 
%drvempls 

